
Table S1: 1H Chemical shifts based on the assignments for the 15 amino acids residues of STEP2 
peptide. 

 

 
No. Residue NH αH βH others 
2 Leu 8.741 4.233 1.928, 2.01  
3 Gln 8.69 4.2566 1.896, 1.981 γCH2 2.32 

δNH2 7.661, 6.968 
4 Glu 8.61 4.2 1.815 γCH2 2.22 

5, 6 Arg 8.62 4.3  γCH2 1.6, 1.741 
δCH2 3.145 
NH 7.23 

7 Gly 8.62 3.9374   
8 Ser 8.381 4.4 3.856, 3.8  
9 Asn 8.645 4.71 2.73, 2.81 γNH2 7.7, 7.0 

10 Val 8.1495 4.088 2.0536 γCH3 0.88 
11 Ser 8.5 4.4 3.8  
12 Leu 8.477 4.37 1.619 γH 1.57 

δCH3 0.825, 0.895 
13 Thr 8.2166 4.3 4.167 γCH3 1.14 
14 Leu 8.34 4.327 1.6 γH 1.55 

δCH3 0.821, 0.887 
15 Asp 8.417 4.55 2.535, 2.695  
16 Met 7.9165 4.23 1.9, 2.05 γCH2 2.44, 2.48 

CH3  

Pulse program for the sequence shown in Fig. 1 of the main text 
 
/* noesyc_purge_lek_600_JA3_cp 
 
Used to record 3D noesy with NOEs between 1H(13C) F1 13C F2 and 1H(12C) F3  
 
Written by LEK on June 21, 2018 for Jacob Anglister 

Intended for 2H protein that is methyl labeled and unlabeled peptide 
 
Purging based on 1H (not 13C) Figure 3a of Zwahlen et al 
 
For 2D F1-F3 set -DC180_flg so that there is no evolution in 13C (t2) 
 
Set f2180 to yes 
 
Recommended usage: -Df2180 -DsepN_flg  
: use 2 different adiabatic purge pulses  
 pwc_ad1 is 2.359 ms, 60kHz sweep, cnst12=30000, cnst13=2.543, taua1=2.2m  
 pwc_ad2 is 1.536 ms,60kHz sweep,cnst14=30000,cnst15=3.905,taua2=2.0m  
 - both pulses are applied with a 5 kHz 13C B1 field 
 
 If you sit at 67 ppm in 13C: 
 pwc_ad1: BRF_apod_chirp 2359 1000 -40110 60000 0 20 0 1 
 pwc_ad2: BRF_apod_chirp 1536 1000 -40110 60000 0 20 0 1 
 
For methyl labeled samples best to sit at 20 ppm 
 If you sit at 20 ppm in 13C: 
 pwc_ad1: BRF_apod_chirp 2359 1000 -33020 60000 0 20 0 1 
 pwc_ad2: BRF_apod_chirp 1536 1000 -33020 60000 0 20 0 1 
 
USE NS=16 
================== 
;;spnam23= chirp_80kHz.BRF  ;; calibrate 22us 90 hard 
;;spnam24= chp_2pt359L.BRF  ;; calibrate 50us 90 hard 
;;spnam25= chp_2pt359L.BRF  ;; calibrate 50us 90 hard 
======================= 
 
This is the preferred version as it works better than 13C purging (Figure 1c) 
Do not need to perform the 1H phase cycling before t1.  
 
Modified by LEK on June 24, 2018 to sit on water 
 
*/ 
 
#include <Avance.incl> 
#include <Grad.incl> 
#include <Delay.incl> 

;Define phases 
#define zero ph=0.0 
#define one ph=90.0 
#define two ph=180.0 
#define three ph=270.0 
  
;Define Pulses 
 
define pulse dly_pg1 
   "dly_pg1=5m" 
define pulse dly_pg2 
   "dly_pg2=dly_pg1/1.62" 
 
define pulse pwh 
       "pwh=p1"             ; 1H hard pulse at power pl1 
define pulse pwc 
       "pwc=p2"             ; 13C hard pulse at power pl2 
define pulse pwc_ad         ; adiabatic pulse for inept transfers 
       "pwc_ad = p23" 
define pulse pwc_ad1        ; adiabatic pulse for first purge 
       "pwc_ad1 = p24" 
define pulse pwc_ad2        ; adiabatic pulse for second purge 
       "pwc_ad2 = p25" 
define pulse pwn 
       "pwn = p3"           ; 15N high power pulse 
 
;Define delays 
 
"in0=inf1/2" 
"in10=inf2/2" 
 
"d11=30m" 
 
define delay taua 
       "taua=d5"            ; < 1/(4JCH) 1.8 ms  
 
define delay NOE_mix 
       "NOE_mix = d6" 
 
define delay taua1 
       "taua1=d7"            ; set according to purge characteristics   
 
define delay taua2 
       "taua2=d8"            ; set according to purge characteristics   
 
define delay d_me1 
   "DELTA=cnst12*(1.0s/cnst13)"   
   "d_me1 = DELTA*(0.0000001)" 

; cnst12 is shift diff between start of sweep and whre carrier is at the time of appln of 1H 180o     
; cnst13 is the rate of sweep in units of 10^7 Hz/s 
 
define delay d_me2 
   "DELTA=cnst14*(1.0s/cnst15)"   
   "d_me2 = DELTA*(0.0000001)" 
 
define delay tau1 
define delay tau2 
 
define delay hscuba 
  "hscuba=30m" 
 
#ifndef OneDarray 
define loopcounter ni 
     "ni = td1/2" 
define loopcounter ni2 
     "ni2 = td2/2" 
#endif 
 
define loopcounter zetacount 
 
;define flags 
;f1180    ; set zgoptns -Df1180 
 
#ifdef f1180 
  "d0=(in0)/2"    
#else 
  "d0=(0.2u)/2" 
#endif 
 
#ifdef f2180 
  "d10=(in10 - pwc*4.0/PI - pwh*4.0)/2" 
#else 
  "d10=(0.2u - pwc*4.0/PI - pwh*4.0)/2" 
#endif 
 
/* Assign cnsts to check validity of parameter changes */ 
#ifdef fsat 
   "cnst10 = plw10" ; tsatpwr - set to max of 0.00005W 
#endif 
 
#ifdef mess_flg 
   "cnst11=plw11" ; power for messerlie purge 
#endif 
 
   "cnst21=plw21" ; dpwr pl21 - set max at 4.5W 

   "cnst23=spw23" ; power level for adiabatic pulses during inept 
   "cnst24=spw24" ; power level for adiabatic pulse for first purge 
   "cnst25=spw25" ; power level for adiabatic pulse for second purge 
 
   "cnst31=plw31" ; dpwr2 pl31 - set max at 5.8W 
 
 
 "acqt0=-6u - pwh*2/PI"  ; select 'DIGIMOD = baseopt' to execute 
 
 
1 ze 
 
; check validity of parameters 
 
;  if "aq > 250m" { 
;   2u 
;   print "error: acquisition time in t3 is too long; < 250 ms" 
;   goto HaltAcqu 
;  } 
 
;#ifdef fsat 
;  if "cnst10 > 0.001" 
;  { 
;   2u 
;   print "error: presat power pl10 is too large" 
;   goto HaltAcqu 
;  } 
;#endif 
 
;#ifdef mess_flg 
;  if "cnst11 > 1.0" 
;  { 
;   2u 
;   print "error: 1H mess_flg power pl11 is too large" 
;   goto HaltAcqu 
;  } 
;#endif 
 
;  if " cnst21 > 7.0" 
;  { 
;   2u 
;   print "error: dpwr pl21 too large" 
;   goto HaltAcqu 
;  } 
 
;  if " cnst23 > 80.0" 
;  { 
;   2u 

;   print "error: power level for adiabatic pules during inept spw23 is too large" 
;   goto HaltAcqu 
;  } 
 
;  if " cnst24 > 20.0" 
;  { 
;   2u 
;   print "error: power level for first purge adiabatic pule spw24 is too large" 
;   goto HaltAcqu 
;  } 
 
;  if " cnst25 > 20.0" 
;  { 
;   2u 
;   print "error: power level for first purge adiabatic pule spw25 is too large" 
;   goto HaltAcqu 
;  } 
 
;  if " cnst31 > 5.8" 
;  { 
;   2u 
;   print "error: dpwr2 pl31 too large" 
;   goto HaltAcqu 
;  } 
 
;  if "pwc > 20u" 
;  { 
;   2u 
;   print "error: pwc too large < 20 us" 
;   goto HaltAcqu 
;  } 
 
;  if "pwc_ad > 500u" 
;  { 
;   2u 
;   print "error: pwc_ad too large < 500 us" 
;   goto HaltAcqu 
;  } 
 
; lock is not applied during acquisition  
6m LOCKH_ON   ; turn on lock hold, ie turn lock off so that it can be toggled back on during first pass 
 
2 d11   
 
6m LOCKH_OFF   ; turn off lock hold, ie turn lock on 
 
#ifdef mess_flg 
 20u pl11:f1         ;   tpwrmess 

 (dly_pg1 zero):f1   ;   Hdly_pg1x 
 20u 
 (dly_pg2 one):f1    ;   Hdly_pg2y 
 20u pl10:f1         ;   tsatpwr 
#endif 
 
#ifdef fsat                           ; zgoptn -Dfsat 
  4u pl10:f1                          ; power(tsatpwr) for presaturation 
  2u fq=0:f1                          ; set 1H carrier on water 
  d1 cw:f1 zero                       ; Hcw(d1)x 
  4u do:f1                            ; cw off 
  2u fq=cnst1:f1                      ; set 1H carrier at methyls 
  2u pl1:f1                           ; power(tpwr) 
 
#ifdef fscuba                   /* Scuba pulse sequence */ 
   hscuba                          ; delay(hscuba) 
    (pwh zero):f1                  ; H 90x180y90x 
    (pwh*2 one):f1 
    (pwh zero):f1 
   hscuba                          ; delay(hscuba) 
#endif                           /* end fscuba */ 
 
#else                             /* if fsat is no */ 
 2u pl1:f1                        ; power(tpwr) 
 d1                               ; delay(d1) 
#endif                            /* end if fsat */ 
 
  2u fq=cnst1:f1         ; set 1H carrier on methyls 
  2u fq=0:f2             ; set 13C carrier at o2 
  2u pl1:f1 
  2u pl2:f2 
  2u pl3:f3 
 
 20u UNBLKGRAD           ; dly 20u, unblank gradients and lock on hold 
 
  (pwc zero):f2   ; C90x - To destroy 13C Boltzman magnetization 
 
  2u  
  (p50:gp0)       ; gradient 0 
  d16 
 
; This is the real start 
 
"tau1 = d0" 
 
"tau2 = d10" 
 
if "tau2 < 0.2u" { 

   "tau2 = 0.2u" 
} 
 
  (pwh ph1):f1                             
 
  2u 
  p51:gp1 
  d16 
 
  "DELTA = tau1 + taua - pwc_ad*0.5 - 2u - p51 - d16"  
  DELTA 
 
  (pwc_ad:sp23 ph26):f2  
 
  tau1 
 
  (pwh*2 ph26):f1 
 
  "DELTA = taua + pwc_ad - pwc_ad*0.5 - 2u - p51 - d16"  
  DELTA 
 
  2u 
  p51:gp1 
  d16 
 
  (pwh ph27):f1                             
 
  2u 
  p52:gp2     ; use a strong gradient to assist with water dephasing 
  d16 
 
  2u pl2:f2   ; set 13C to high power 
 
  (pwc ph2):f2 
 
#ifndef C180_flg 
   tau2 
   (pwh ph26 pwh*2 ph27 pwh ph26):f1 
   tau2 
#else 
  (center(pwh ph27 pwh*2 ph26 pwh ph27):f1 (pwc ph26 pwc*2 ph27 pwc ph26):f2)   
#endif 
 
  (pwc ph26):f2 
 
  2u 
  p53:gp3     ; use a strong gradient to assist with water dephasing gp2*gp3 < 0 
  d16 

  (pwh ph26):f1 
 
  2u 
  p54:gp4                       ; gradient 4 
  d16 
 
 "DELTA = taua - 2.0u - p54 - d16 - pwc_ad*0.5" 
  DELTA                                    ; delay 1/4JCH 
 
  (center(pwh ph27 2u pwh*2 ph26 2u pwh ph27):f1 (pwc_ad:sp23 ph26):f2)   
  2u pl2:f2 
 
 "DELTA = taua - 2u - p54 - d16 - pwc_ad*0.5 - 2u" 
  DELTA                                    ; delay 1/4JCH 
 
  2u 
  p54:gp4                                 ; gradient 4 
  d16 
 
  (pwh ph27):f1                             
 
#ifndef sat_noe 
 
  "DELTA = NOE_mix - 10m" 
   DELTA 
 
  2u 
  p55:gp5                                  ; gradient 5 
  d16 
 
  2u pl2:f2 
  (pwc zero):f2 
 
  2u 
  p56:gp6                                  ; gradient 6 
  d16 
 
  "DELTA = 10m - 2u - p55 - d16 - 2u - pwc - 2u - p56 - d16" 
   DELTA 
 
#else 
  
  2u 
  p55:gp5                                  ; gradient 5 
  d16 
 
  2u pl2:f2 

  (pwc zero):f2 
 
  2u 
  p56:gp6                                  ; gradient 6 
  d16 
 
  "DELTA = NOE_mix - 2u - p55 - d16 - 2u - pwc - 2u - p56 - d16 - 2u - 2u - 4u - 2u - 2u"  
 
  2u pl10:f1                          ; power(tsatpwr) for presaturation 
  2u fq=0:f1                          ; set 1H carrier on water 
  DELTA cw:f1 zero                    ; Hcw(d1)x 
  4u do:f1                            ; cw off 
  2u fq=cnst1:f1                      ; set 1H carrier on methyls 
  2u pl1:f1                           ; power(tpwr) 
 
#endif 
 
; end of mixing time begin filtering, potential for 15N filtering if desired; else just set power down 
 (pwh ph3):f1                              
 
  2u 
  (p57:gp7)       ; gradient 7 
  d16 
 
  "DELTA = taua1 - 2u - p57 - d16 - d_me1" 
   DELTA 
 
   "DELTA = d_me1 - (pwn - pwh)" 
 
   (lalign(d_me1 pwh*2 ph26):f1 (pwc_ad1:sp24 ph26):f2 (DELTA pwn*2 ph26):f3) 
 
   "DELTA = taua1 - pwc_ad1 + d_me1 + pwh*2.0 - 2u - p57 - d16" 
    DELTA 
 
  2u 
  (p57:gp7)       ; gradient 7 
  d16 
 
   (pwh zero):f1 
 
  2u 
  (p60:gp10)       ; gradient 10 
  d16 
 
  (pwh ph4):f1     ; second purge step 
 
  2u 
  (p58:gp8)       ; gradient 8 

  d16 
 
  "DELTA = taua2 - 2u - p58 - d16 - d_me2" 
   DELTA 
 
   "DELTA = d_me2 - (pwn - pwh)" 
 
   (lalign(d_me2 pwh*2 ph26):f1 (pwc_ad2:sp25 ph26):f2 (DELTA pwn*2 ph26):f3) 
 
   "DELTA = taua2 - pwc_ad2 + d_me2 + pwh*2.0 - 2u - p58 - d16" 
    DELTA 
 
  2u 
  (p58:gp8)       ; gradient 8 
  d16 
 
  (pwh zero):f1 
 
  2u 
  (p61:gp11)       ; gradient 11 
  d16 
 
  2u fq=0:f1                      ; set 1H carrier on water 
  2u pl1:f1 
  2u pl2:f2 
  2u pl3:f3 
 
  (pwh ph26):f1 
 
  4u BLKGRAMP   ; turn grad amplifiers off but keep lock off as well 
 
  2u pl21:f2               ; IMPORTANT!!  lower power for 13C decoupling 
  2u pl31:f3               ; IMPORTANT!!  lower power for 15N decoupling 
 
  go=2 ph31  
  d11  mc #0 to 2 
  F1PH(calph(ph1, +90),caldel(d0, +in0)) 
  F2PH(calph(ph2, +90),caldel(d10, +in10) & calph(ph2, +180) & calph(ph31, +180)) 
 
6m LOCKH_OFF   ; turn off lock hold, ie turn lock on 
 
HaltAcqu, 1m 
exit 
 
ph1 = 0 0 0 0 2 2 2 2  ; 1H pulse before t1 
ph2 = 0 0 0 0 0 0 0 0 2 2 2 2 2 2 2 2 ; 13C pulse before t2 
ph3=0 2 ; 1H purge 1 
ph4=0 0 2 2      ; 1H second purge 

ph31=0 2 2 0 2 0 0 2 2 0 0 2 0 2 2 0 
ph26=0 
ph27=1 
ph28=2 
ph29=3 
 
;d1 : repetition delay 
;d5 : taua ~1/4JCH 
;d6 : mixing time for NOE 
;d7 : taua1 = time for first inept purge 
;d8 : taua2 = time for second inept purge 
;d11 : delay for disk i/o, 30ms 
;d16 : gradient recovery delay, 200us 
;pl1 : tpwr - power level for pwh  
;pl2 : dhpwr - power level for hard 13C pulse pwc  
;pl21 : dpwr - power level for  13C decoupling cpd2 
;sp23 : power level for pwc_ad (inept transfers) 
;sp24 : power for for pwc_ad1 (first purge) 
;sp25 : power for for pwc_ad2 (second purge) 
;p23  : pwc_ad - adiabatic pulses for INEPT transfer 
;p24  : pwc_ad1 - adiabatic pulse for first purge 
;p25  : pwc_ad2 - adiabatic pulses for second purge 
;pl31 : dpwr2 - power level for 15N cpd3 
;p1 : pwh  
;p2 : pwc   
;p3  : pwn  
;p31 : pwn at dpwr2 for 15N decoupling during t1 peried and t3 > 300 us Set to 1000 for N_off 
;ni : td1/2 number of complex points in t1 
;ni2 : td2/2 number of complex points in t2 
;cpd2 : 13C decoupling according to program defined by cpdprg2 
;cpd3 : 15N decoupling according to program defined by cpdprg3 
;pcpd2:  13C 90 degree pulse at pl21 for cpd2 
;pcpd3:  15N 90 degree pulse at pl31 for cpd3 
;spnam23 : file name for chirp pulse during inepts 
;spnam24 : file name for first chirp purge pulse  
;spnam25 : file name for second chirp purge pulse  
;cnst1  : methyls - water < 0  in Hz 
;cnst21 : power level for 13C decoupling 
;cnst23 : power level for adiabatic pulse during inept transfers (pwc_ad) 
;cnst24 : power level for first adiabatic purge (pwc_ad1) 
;cnst25 : power level for second adiabatic purge (pwc_ad2) 
;cnst31 : power level for 15N decoupling during t1 and t3 
;cnst12 : Hz between start of first adiabatic and its position at time of 1H 180 
;cnst13 : sweep rate of first adiabatic in units of 10^7Hz/s 
;cnst14 : Hz between start of second adiabatic and its position at time of 1H 180 
;cnst15 : sweep rate of second adiabatic in units of 10^7Hz/s 
;zgoptns : Df1180,Df2180,Dfsat,DOneDarray,Dfscuba,DC180_flg,Dmess_flg,Dsat_noe 
 


